# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xlsclients
pkgver=1.1.4
pkgrel=1
pkgdesc="List client applications running on a display"
arch=('x86_64')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxcb')
makedepends=('xorg-util-macros')
groups=('xorg-apps' 'xorg')
source=(https://xorg.freedesktop.org/archive/individual/app/xlsclients-${pkgver}.tar.bz2{,.sig})
sha512sums=('42fd848d88d3cf31168a37fed63f9f5b60ddc4f36be2956b354e574c93129d75f29e005d6631784380bcef27860b54c64f419ca18ad77f092f4520dec47b10f4'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd xlsclients-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd xlsclients-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
